```
> restart;
> Production := {q = g[1]*L-g[2]*L*L};
```
$$Production := \{q = g_1 L - g_2 L^2\} \quad (1)$$

```
> Marginal_Product := {MPL = diff(subs(Production,q),L)};
```
$$Marginal\_Product := \{MPL = g_1 - 2 g_2 L\} \quad (2)$$

```
> Revenue := {R=p*q};
```
$$Revenue := \{R = p\, q\} \quad (3)$$

```
> MarginalRevenueProduct := {MRPL=diff(subs(Revenue,Production,R),
L)};
```
$$MarginalRevenueProduct := \{MRPL = p\,(g_1 - 2 g_2 L)\} \quad (4)$$

```
>  {w=MRPL};
```
$$\{w = MRPL\} \quad (5)$$

```
> LaborDemand[firm] := subs(MarginalRevenueProduct,%);
```
$$LaborDemand_{firm} := \{w = p\,(g_1 - 2 g_2 L)\} \quad (6)$$

```
> employment[firm] := subs(L=L[f],solve(%,{L}));
```
$$employment_{firm} := \left\{L_f = \frac{1}{2}\,\frac{-w + p\, g_1}{p\, g_2}\right\} \quad (7)$$

```
> Employment[industry] := { L = N*L[f]};
```
$$Employment_{industry} := \{L = N L_f\} \quad (8)$$

```
> subs(employment[firm],Employment[industry]);
```
$$\left\{L = \frac{1}{2}\,\frac{N\,(-w + p\, g_1)}{p\, g_2}\right\} \quad (9)$$

```
> LaborDemand[industry] := solve(%,{w});
```
$$LaborDemand_{industry} := \left\{w = -\frac{(2 g_2 L - N g_1)\, p}{N}\right\} \quad (10)$$

```
> data := {N=100,p=10,a[m]=80,b[m]=1/40,a[f]=-20,b[f]=1/10, g[1]=
30,g[2]=1/10};
```
$$data := \left\{N = 100,\ p = 10,\ a_m = 80,\ b_m = \frac{1}{40},\ a_f = -20,\ b_f = \frac{1}{10},\ g_1 = 30,\ g_2 = \frac{1}{10}\right\} \quad (11)$$

```
> subs(data,LaborDemand[industry]);
```
$$\left\{w = -\frac{1}{50} L + 300\right\} \quad (12)$$

```
> Supply[men] := {w[m]=a[m]+b[m]*L[m]};
```
$$Supply_{men} := \{w_m = a_m + b_m L_m\} \quad (13)$$

```
> Supply[women] := {w[f]=a[f]+b[f]*L[f]};
```
$$Supply_{women} := \{w_f = a_f + b_f L_f\} \quad (14)$$

```
> Supply[both] := {L=L[f]+L[m]};
```
(15)

$$Supply_{both} := \{L = L_f + L_m\} \quad \mathbf{(15)}$$

```
> Employment[men] := solve(subs(w[m]=w,Supply[men]),{L[m]});
```

$$Employment_{men} := \left\{L_m = \frac{w - a_m}{b_m}\right\} \quad \mathbf{(16)}$$

```
> Employment[women] := solve(subs(w[f]=w,Supply[women]),{L[f]});
```

$$Employment_{women} := \left\{L_f = \frac{w - a_f}{b_f}\right\} \quad \mathbf{(17)}$$

```
> LaborSupply[industry] := solve(subs(Employment[men],Employment
  [women],Supply[both]),{w});
```

$$LaborSupply_{industry} := \left\{w = \frac{L\, b_f\, b_m + a_f\, b_m + a_m\, b_f}{b_m + b_f}\right\} \quad \mathbf{(18)}$$

```
> subs(data,LaborSupply[industry]);
```

$$\left\{w = \frac{1}{50}\, L + 60\right\} \quad \mathbf{(19)}$$

```
> Equilibrium := solve(`union`(LaborDemand[industry],LaborSupply
  [industry]),{w,L});
```

$$Equilibrium := \left\{L = \frac{N\,(b_m\, p\, g_1 - a_f\, b_m - a_m\, b_f + b_f\, p\, g_1)}{N\, b_f\, b_m + 2\, b_m\, p\, g_2 + 2\, b_f\, p\, g_2}, w = \frac{p\,(2\, g_2\, a_f\, b_m + 2\, g_2\, a_m\, b_f + g_1\, N\, b_f\, b_m)}{N\, b_f\, b_m + 2\, b_m\, p\, g_2 + 2\, b_f\, p\, g_2}\right\} \quad \mathbf{(20)}$$

```
> subs(data, Equilibrium);
```

$$\{w = 180, L = 6000\} \quad \mathbf{(21)}$$

```
> plots[implicitplot](subs(data,LaborSupply[industry]),L=3000.
  .10000,w=100..300,title="Supply",thickness=3, font=[Helvetica,
  bold,14],titlefont=[Helvetica,bold,16]);
> ps := %:
```




```
> plots[implicitplot](subs(data,LaborDemand[industry]),L=3000.
  .10000,w=100..300,title="Demand",thickness=3,font=[Helvetica,
  bold,14],titlefont=[Helvetica,bold,16]);
> pd := %:
```




```
> plots[display]([ps,pd],title="Equilibrium");
```




```
>
>
>
```

Labor Supply Details

```
>
> Supply[both] := {L=max(L[f],0)+max(L[m],0)};
```

$$Supply_{both} := \{L = \max(0, L_f) + \max(0, L_m)\} \quad (1.1)$$

```
> subs(Employment[men],Employment[women],data,Supply[both]);
> plots[implicitplot](%,L=000..8000,w=0..300,title="Supply",
  thickness=3, font=[Helvetica,bold,14],titlefont=[Helvetica,
  bold,16],numpoints=50000);
> ps2 := %:
```

$$\{L = \max(0, 10\,w + 200) + \max(0, 40\,w - 3200)\}$$




```
> plots[display]([ps2,pd],title="Equilibrium");
```


Equilibrium
200
150
w
100
50
0
1,000
3,000
5,000
7,000
10,000
L

>

>

>